
	Reason for change:
	In these test cases while generating a downlink PDCP PDU, in function  f_GetN_OctetsFromPRBS() we are appending one octet at a time from tsc_RandomPRBS  to v_Noctets. For higher category UEs the octet string length (p_N) will be very large. It has been observed (dependent on PC speed and TTCN-3 compiler implementation) that this loop implementation can cause the total test case time to exceed the currently configured guard timer.


	
	

	Summary of change:
	A more efficient implementation of  f_GetN_OctetsFromPRBS() is proposed to reduce the number of concatenations by appending larger segments of tsc_RandomPRBS  (up to the full length of 2047 bytes) in order to reduce the overall test case execution time.

Change 1
	Function name
	f_GetN_OctetsFromPRBS

	Reason for change
	In these test cases while generating a downlink PDCP PDU, in function  f_GetN_OctetsFromPRBS() we are appending one octet at a time from tsc_RandomPRBS  to v_Noctets. For higher category UEs the octet string length (p_N) will be very large. It has been observed (dependent on PC speed and TTCN-3 compiler implementation) that this loop implementation can cause the total test case time to exceed the currently configured guard timer.


	Summary of change
	A more efficient implementation of  f_GetN_OctetsFromPRBS() is proposed to reduce the number of concatenations by appending larger segments of tsc_RandomPRBS  (up to the full length of 2047 bytes) in order to reduce the overall test case execution time.



	TTCN module
	CommonDefs.ttcn

	MCC160 Comment
	Accepted as improvement with simplified implementation (see below)


Before change
function f_GetN_OctetsFromPRBS(UInt_Type p_Start,

                                 UInt_Type p_N)

    return octetstring

  {

    var integer i;

    var integer k;

    var octetstring v_Noctets :=''O; // output string to be extracted from random test pattern

    for (i:= 0; i< p_N; i := i+1) {

      k := p_Start + i;

      k := k mod 2047;

      v_Noctets := v_Noctets & tsc_RandomPRBS[k];

    }

    return v_Noctets;
  }
After change
  function f_GetN_OctetsFromPRBS(UInt_Type p_Start,

                                 UInt_Type p_N)

    return octetstring

  {

    const integer tsc_RandomPRBSlength:=lengthof(tsc_RandomPRBS);  // length is 2047 with the current value

    var integer i,copied,noFullStrings;

    var integer k;

    var octetstring v_Noctets :=''O; // output string to be extracted from random test pattern

    var integer possibleSubstringLength;

    possibleSubstringLength:=tsc_RandomPRBSlength-p_Start;  // maximal length that can be copied in first step

    if (p_N<possibleSubstringLength) {  // result can be determined in one step

      v_Noctets:=substr(tsc_RandomPRBS,p_Start,p_N);

    }

    else { //  result determined in multiple steps

      copied:=tsc_RandomPRBSlength-p_Start;

      v_Noctets:=substr(tsc_RandomPRBS,p_Start,copied); // same as first step above

      p_N:=p_N-copied;

      noFullStrings:=p_N/tsc_RandomPRBSlength; // no of total buffer to copy

      for (i:=0;i<noFullStrings;i:=i+1) {

        v_Noctets:=v_Noctets & substr(tsc_RandomPRBS,0,tsc_RandomPRBSlength);  // copy complete buffer

        p_N:=p_N-tsc_RandomPRBSlength;

      }

      if (p_N>0) {

        v_Noctets:=v_Noctets & substr(tsc_RandomPRBS,0,p_N); // copy remaining octets

      }

    }

    return v_Noctets;
  }    

MCC160 Implementation
	  function f_GetN_OctetsFromPRBS(UInt_Type p_Start,

                                 UInt_Type p_N)

    return octetstring

  {

    var integer v_LengthOfPRBS := lengthof(tsc_RandomPRBS);

    var integer v_Start := p_Start mod v_LengthOfPRBS;
    var octetstring v_Noctets := substr(tsc_RandomPRBS, v_Start, v_LengthOfPRBS - v_Start);

    while (lengthof(v_Noctets) < p_N) {

      v_Noctets := v_Noctets & tsc_RandomPRBS;

    }

    v_Noctets := substr(v_Noctets, 0, p_N);

    return v_Noctets;

  }


